MCF Calculations and Example Spreadsheet

Andrew VanderZaag

May 15, 2018

This Annex was developed to explain how MCF factors in the guidelines have been derived and to provide a detailed step by step protocol for inventory compilers to calculate country or region specific MCFs.

Further, to support the IPCC Guidance Document, a spreadsheet was created to enable users to calculate a site-specific Methane Conversion Factor (MCF). The spreadsheet uses the same calculations that were used to calculated the MCF Table in the guidance document, but has been designed with a user in mind and can be provided on request (for review)

The calculation procedure outlined in the spreadsheet contains three main sections:

1. **Inputs** to the model
2. **Model** calculations
3. **Results** from the model

As an explanation of procedures, within each section, cells are colour coded. Compilers are required to develop input data for anything that is indicated by yellow highlighted cells, and have the option of editing the orange highlighted cells if needed, but only if country-specific information is available for those parameters. Other cells are not meant to be edited by the user.

COLOUR CODE:	REQUIRED USER INPUT	OPTIONAL USER INPUT	FIXED INPUT DON'T EDIT	CALCULATION DON'T EDIT

Figure 1. Colour code for cells in the example spreadsheet.

MODEL INPUT

The Input required to recreate the spreadsheet is shown below (Figures 2 and 3). In this section, the compiler should input 12 months of temperature data (degrees C) in cells D9:D20, based on average monthly temperatures for the region for which they wish to develop the MCF.

If the compiler has <u>manure temperature</u> available, they should select “Manure” in cell D6. As a result, the spreadsheet will copy the user-input temperature into cells E9:E20, for further use in the analysis.

If the compiler has <u>air temperature</u> (not manure temperature), they should select “Air” in cell D6. As a result, the spreadsheet will generate an estimate for manure temperature in cells E9:E20. The estimates are based on the following logic:

- Manure temperature lags 1-month behind air temperature.
  - e.g., Tmanure in June = Tair in May.
- The minimum manure temperature will be used (1 degree C by default; user adjustable)
  - e.g., for Tair = -9 C, Tman = 1 C
- If and only if the storage is emptied once per year, manure temperature will be reduced by a dampening factor (3 degrees C by default; user adjustable).
  - i.e. Tman = Tair – damping factor; e.g., 12 = 15 – 3
- The logic equation is implemented in Excel as follows, for example, in cell E9:
  =IF($D$6="Manure",D9,IF($F$21>1,MAX(D20,f_Tmin),MAX(D20-f_T2damping,f_Tmin)))
  - Broken into steps:
  - If $D$6="Manure" then the result in E9 will equal D9
  - If $D$6 is not "Manure" (i.e. it is “Air”) then the second IF statement is operated
    - IF $F$21>1 (i.e. multiple removals per year), then no damping is applied

        - Manure temperature is selected as air temperature from the previous month, and it is always greater or equal to the minimum temperature,
          i.e. E9 will equal MAX(D20,f_Tmin). In this case, D20 (-6.7) is less than the minimum, so the result in E9 is the minimum (1.0).
    - IF $F$21=1 then damping is applied
        - Damping is applied by subtracting the damping factor: D20-f_T2damping
        - The temperature is always greater or equal to the minimum temperature, using the MAX() function.

The compiler should then identify the months when manure is removed from the storage in column F (F9:F20). This can be indicated by a "Y" indicating months when manure was removed, and an "N" for months when manure is not removed. The number of months when manure was removed is counted and displayed in cell F21.

	A	B	C	D	E	F
4						
5		**INPUTS:**				
6				Air	Manure	
7				Temperature	Temperature	Manure
8		**Month**	**Month**	**°C**	**°C**	Removed (Y/N)
9		January	**1**	-9.0	1.0	N
10		February	**2**	-7.7	1.0	N
11		March	**3**	-2.3	1.0	N
12		April	**4**	4.7	1.0	N
13		May	**5**	10.7	4.7	Y
14		June	**6**	15.2	10.7	N
15		July	**7**	17.7	15.2	N
16		August	**8**	16.7	17.7	N
17		September	**9**	12.0	16.7	N
18		October	**10**	5.8	12.0	N
19		November	**11**	-1.4	5.8	Y
20		December	**12**	-6.7	1.0	N
21				4.6	7.3	2
22				*Average*	*Average*	*Count of "Y"*



	A	B	C	D	E	F
4						
5		**INPUTS:**				
6				Air	Manure	
7				Temperature	Temperature	Manure
8		**Month**	**Month**	°C	°C	Removed (Y/N)
9		January	1	-9	=IF($D$6="Manure",D9,IF($F$21>1,MAX(D20,f_Tmin),MAX(D20-f_T2damping,f_Tmin)))	N
10		February	2	-7.7	=IF($D$6="Manure",D10,IF($F$21>1,MAX(D9,f_Tmin),MAX(D9-f_T2damping,f_Tmin)))	N
11		March	3	-2.3	=IF($D$6="Manure",D11,IF($F$21>1,MAX(D10,f_Tmin),MAX(D10-f_T2damping,f_Tmin)))	N
12		April	4	4.7	=IF($D$6="Manure",D12,IF($F$21>1,MAX(D11,f_Tmin),MAX(D11-f_T2damping,f_Tmin)))	N
13		May	5	10.7	=IF($D$6="Manure",D13,IF($F$21>1,MAX(D12,f_Tmin),MAX(D12-f_T2damping,f_Tmin)))	Y
14		June	6	15.2	=IF($D$6="Manure",D14,IF($F$21>1,MAX(D13,f_Tmin),MAX(D13-f_T2damping,f_Tmin)))	N
15		July	7	17.7	=IF($D$6="Manure",D15,IF($F$21>1,MAX(D14,f_Tmin),MAX(D14-f_T2damping,f_Tmin)))	N
16		August	8	16.7	=IF($D$6="Manure",D16,IF($F$21>1,MAX(D15,f_Tmin),MAX(D15-f_T2damping,f_Tmin)))	N
17		September	9	12	=IF($D$6="Manure",D17,IF($F$21>1,MAX(D16,f_Tmin),MAX(D16-f_T2damping,f_Tmin)))	N
18		October	10	5.8	=IF($D$6="Manure",D18,IF($F$21>1,MAX(D17,f_Tmin),MAX(D17-f_T2damping,f_Tmin)))	N
19		November	11	-1.4	=IF($D$6="Manure",D19,IF($F$21>1,MAX(D18,f_Tmin),MAX(D18-f_T2damping,f_Tmin)))	Y
20		December	12	-6.7	=IF($D$6="Manure",D20,IF($F$21>1,MAX(D19,f_Tmin),MAX(D19-f_T2damping,f_Tmin)))	N
21				=AVERAGE(D9:D20)	=AVERAGE(E9:E20)	=COUNTIF(F9:F20,"Y")
22				*Average*	*Average*	*Count of "Y"*

Figure 2. Temperature and manure removal inputs to the model. Top panel: alphanumeric values in each cell. Middle panel: dropdown menu to select "Air" or "Manure". Bottom panel: all formulae are visible.

The inventory compiler is required to provide several other inputs in the section shown below (Fig. 3). The name of each parameter is provided in column H, the numeric value of the parameter is in column I, the units are in column J, the source each value are given in column K, additional notes are in columns L and M, default values are in column N. To make equations more easily understood, the Microsoft Excel feature of "Named Cells" has been used to name the cells in column I, and the name of each cell is shown in column O for convenience. For example, cell I7 is given the name "VS_PROD_YR". See Figure 3 for a full list of named cells.

Additional information about the input parameters:

- VS Excretion – based on IPCC guidance.
- VS % liquid storage – this indicates what percentage of excreted VS is handled as a liquid. For example,
    - 100% indicates that all excreted VS enters the liquid storage
    - A lower number (say, 75%) could indicate that a portion of the solids are separated by a screwpress and handled as a solid (25%) while the remaining 75% is handled as liquid.
- The compiler must provide a B_0 value for the manure. Refer to IPCC guidance.
- The compiler may, optionally, adjust the minimum temperature (and temperature damping factors.
- The compiler also has the option to adjust the emptying efficiency, which indicates the percentage of manure removed from storage at each removal. By default this is set to 95%, indicating that 5% of the VS remain in storage after emptying. Set this value to 100% for complete removal.



Parameter:	Value	Units	Source:	Notes	Default Value:	Named Cell
VS Excretion:	1200	kg/year	user input	based on VS excretion, and manure handling	n/a	VS_PROD_YR
VS % liquid storage:	100%	%	user input	% going to liquid storage (e.g. solid-liquid separat	100%	VS_PCT_LIQUID
				note emissions from solid must be handled separately		
f equation:	f = EXP((Ea*(T2-T1))/(R*T2*303.16))					
T1	308.16	K	Temperature of B0 assays		308.16	f_T1
T2	monthly input	K	user input	Enter in Column D (D9:D20)	n/a	n/a
Ea	19347	cal/mol	Petersen et al. PLoS One	(compare: 15175 from Mangino et al. 2001)	19347	f_Ea
R	1.987	cal/K.mol	Mangino et al. 2001		1.987	f_R
Minimum T2	1.0	C	Judgement.	converted to K in calculation	1.0	f_Tmin
Damping T2	3.0	C	Judgement (Rennie et al. 2017.)	applied only when manure removed once per year		f_T2damping
B0	0.24	m3/kg VS added	user input	refer to IPCC guidance for default B0 values	n/a	B0
MDP	1.0	unitless	MDP is not used (i.e. =1.0). Adjust VS % liquid storage or excretion instead.		1.0	
emptying efficiency	95%	%	Judgement. Default 95%	Percent of manure removed (1-residual)	95%	EMPTY_EFFICIENCY
CH4 density	0.662	kg/m3	IPCC guidance		0.662	CH4_DENSITY

Name Manager

Name	Value	Refers To	Scope	Comment
B0	0.24	='MCF Model'!$I$17	Workbook	
CH4_DENSITY	0.662	='MCF Model'!$I$21	Workbook	
EMPTY_EFFICIE...	95%	='MCF Model'!$I$20	Workbook	
f_Ea	19347	='MCF Model'!$I$13	Workbook	
f_R	1.987	='MCF Model'!$I$14	Workbook	
f_T1	308.16	='MCF Model'!$I$11	Workbook	
f_T2damping	3.0	='MCF Model'!$I$16	Workbook	
f_Tmin	1.0	='MCF Model'!$I$15	Workbook	
VS_PCT_LIQUID	100%	='MCF Model'!$I$8	Workbook	
VS_PROD_YR	1200	='MCF Model'!$I$7	Workbook	

Refers to: ='MCF Model'!$I$17

Figure 3. Constants and other input parameters for the model are shown in the top panel. Named Cells in column I are shown in column O, and in the Name Manager dialog box (bottom panel). No formulae exist in this part of the spreadsheet.

MODEL CALCULATIONS

The model calculations are run for three years, in order to ensure VS available has stabilized on an annual basis. For example, in Figure 4, we see that VS Available (column J) increases substantially from the first year to the second year (J64 vs J65), and then stabilizes in the third year (J66). This is because the first year begins from a perfectly empty storage, whereas the second year is emptied according to the Emptying Efficiency parameter (95% removed / 5% remaining; Figure 3).

The model approach is as follows:

- Column B: Month of year, over 3 years. These month numbers are used to extract input data shown in Figure 2.
- Column C: Average manure temperature in each month. This is extracted from cells E9:E20 (Fig. 2) using a VLOOKUP function (Figure 5).
- Column D: temperature is converted from Celsius to Kelvin, using Excel's CONVERT function (Fig. 5).
- Column E: the temperature-dependent *f* parameter is calculated using the van't Hoff-Arrhenius equation (Mangino et al. 2001; IPCC 2006), with updated input parameters shown in Figure 3.
- Column F: monthly VS excreted is calculated by dividing the annual VS input parameter by 12.
- Column G: monthly VS loaded is calculated by multiplying VS excreted by the percentage stored as liquid. In this example, the two are equal because VS_PCT_LIQUID is 100% (Fig. 3).
- Column H: monthly manure emptying is extracted from cells F9:F20 (Fig. 2) using a VLOOKUP function (Fig. 5).
- Column I: the quantity of VS emptied is calculated. The logic is as follows: if emptying occurred, then calculate the mass of VS available to be removed using the mass of VS available in the previous month minus the mass of VS consumed in the previous month. Then, multiply the result by the EMPTY_EFFICIENCY parameter (Fig. 3, 5).
- Column J: the mass of VS available for producing methane is calculated. In the first month of the first year this is equal to the mass of VS loaded. In all other months, this is calculated as the VS loaded in the current month + VS available in the previous month – VS consumed in the previous month – VS emptied in the current month.
- Column K: the mass of VS consumed is calculated by multiplying VS available by *f*.
- Column L: the volume of CH4 produced is calculated by multiplying VS consumed by B_0.

Using these values and equations, the compiler should be able to reproduce graphics such as the profile of manure temperature, volatile solids and methane production shown in Figure 6.

MODEL:

Month	(lookup) Tav_C	(converted) Tav_K	(calculated) *f*	*VS Excreted kg/month*	*VS Loaded kg/month*	Emptying *Y/N*	VS Emptied *kg*	VS "Available" *kg*	VS "consumed" *kg*	CH4 Produced *m3*
1	1.0	274.15	0.02	100	100	N	n/a	100	2	0
2	1.0	274.15	0.02	100	100	N	-	198	4	1
3	1.0	274.15	0.02	100	100	N	-	294	6	1
4	1.0	274.15	0.02	100	100	N	-	388	8	2
5	4.7	277.85	0.03	100	100	Y	362	119	4	1
6	10.7	283.85	0.07	100	100	N	-	215	14	3
7	15.2	288.35	0.11	100	100	N	-	301	34	8
8	17.7	290.85	0.15	100	100	N	-	367	56	13
9	16.7	289.85	0.14	100	100	N	-	411	56	13
10	12.0	285.15	0.08	100	100	N	-	455	36	9
11	5.8	278.95	0.04	100	100	Y	398	121	4	1
12	1.0	274.15	0.02	100	100	N	-	217	4	1
1	1.0	274.15	0.02	100	100	N	-	312	6	1
2	1.0	274.15	0.02	100	100	N	-	406	8	2
3	1.0	274.15	0.02	100	100	N	-	498	10	2
4	1.0	274.15	0.02	100	100	N	-	588	12	3
5	4.7	277.85	0.03	100	100	Y	548	129	4	1
6	10.7	283.85	0.07	100	100	N	-	225	15	4
7	15.2	288.35	0.11	100	100	N	-	310	35	8
8	17.7	290.85	0.15	100	100	N	-	374	57	14
9	16.7	289.85	0.14	100	100	N	-	417	57	14
10	12.0	285.15	0.08	100	100	N	-	461	36	9
11	5.8	278.95	0.04	100	100	Y	403	121	4	1
12	1.0	274.15	0.02	100	100	N	-	217	4	1
1	1.0	274.15	0.02	100	100	N	-	312	6	1
2	1.0	274.15	0.02	100	100	N	-	406	8	2
3	1.0	274.15	0.02	100	100	N	-	498	10	2
4	1.0	274.15	0.02	100	100	N	-	588	12	3
5	4.7	277.85	0.03	100	100	Y	548	129	4	1
6	10.7	283.85	0.07	100	100	N	-	225	15	4
7	15.2	288.35	0.11	100	100	N	-	310	35	8
8	17.7	290.85	0.15	100	100	N	-	374	57	14
9	16.7	289.85	0.14	100	100	N	-	417	57	14
10	12.0	285.15	0.08	100	100	N	-	461	36	9
11	5.8	278.95	0.04	100	100	Y	403	121	4	1
12	1.0	274.15	0.02	100	100	N	-	217	4	1
			SUM:							
			Year 1	1,200	1,200		760	3,185	228	55
			Year 2	1,200	1,200		951	4,058	249	60
			Year 3	1,200	1,200		951	4,059	249	60

Figure 4. Model inputs and outputs over a three year period.

	A	B	C	D	E	F	G
23							
24		**MODEL:**					
25			*(lookup)*	*(converted)*	*(calculated)*	*VS Excreted*	*VS Loaded*
26		**Month**	**Tav_C**	**Tav_K**	***f***	***kg/month***	***kg/month***
27		**1**	=VLOOKUP(B27,$C$9:$E$20,3,FALSE)	=CONVERT(C27,"C","K")	=EXP((f_Ea*(D27-f_T1))/(f_R*D27*f_T1))	=VS_PROD_YR/12	=F27*VS_PCT_LIQUID
28		**2**	=VLOOKUP(B28,$C$9:$E$20,3,FALSE)	=CONVERT(C28,"C","K")	=EXP((f_Ea*(D28-f_T1))/(f_R*D28*f_T1))	=VS_PROD_YR/12	=F28*VS_PCT_LIQUID
29		**3**	=VLOOKUP(B29,$C$9:$E$20,3,FALSE)	=CONVERT(C29,"C","K")	=EXP((f_Ea*(D29-f_T1))/(f_R*D29*f_T1))	=VS_PROD_YR/12	=F29*VS_PCT_LIQUID
30		**4**	=VLOOKUP(B30,$C$9:$E$20,3,FALSE)	=CONVERT(C30,"C","K")	=EXP((f_Ea*(D30-f_T1))/(f_R*D30*f_T1))	=VS_PROD_YR/12	=F30*VS_PCT_LIQUID
31		**5**	=VLOOKUP(B31,$C$9:$E$20,3,FALSE)	=CONVERT(C31,"C","K")	=EXP((f_Ea*(D31-f_T1))/(f_R*D31*f_T1))	=VS_PROD_YR/12	=F31*VS_PCT_LIQUID
32		**6**	=VLOOKUP(B32,$C$9:$E$20,3,FALSE)	=CONVERT(C32,"C","K")	=EXP((f_Ea*(D32-f_T1))/(f_R*D32*f_T1))	=VS_PROD_YR/12	=F32*VS_PCT_LIQUID
33		**7**	=VLOOKUP(B33,$C$9:$E$20,3,FALSE)	=CONVERT(C33,"C","K")	=EXP((f_Ea*(D33-f_T1))/(f_R*D33*f_T1))	=VS_PROD_YR/12	=F33*VS_PCT_LIQUID
34		**8**	=VLOOKUP(B34,$C$9:$E$20,3,FALSE)	=CONVERT(C34,"C","K")	=EXP((f_Ea*(D34-f_T1))/(f_R*D34*f_T1))	=VS_PROD_YR/12	=F34*VS_PCT_LIQUID
35		**9**	=VLOOKUP(B35,$C$9:$E$20,3,FALSE)	=CONVERT(C35,"C","K")	=EXP((f_Ea*(D35-f_T1))/(f_R*D35*f_T1))	=VS_PROD_YR/12	=F35*VS_PCT_LIQUID
36		**10**	=VLOOKUP(B36,$C$9:$E$20,3,FALSE)	=CONVERT(C36,"C","K")	=EXP((f_Ea*(D36-f_T1))/(f_R*D36*f_T1))	=VS_PROD_YR/12	=F36*VS_PCT_LIQUID
37		**11**	=VLOOKUP(B37,$C$9:$E$20,3,FALSE)	=CONVERT(C37,"C","K")	=EXP((f_Ea*(D37-f_T1))/(f_R*D37*f_T1))	=VS_PROD_YR/12	=F37*VS_PCT_LIQUID
38		**12**	=VLOOKUP(B38,$C$9:$E$20,3,FALSE)	=CONVERT(C38,"C","K")	=EXP((f_Ea*(D38-f_T1))/(f_R*D38*f_T1))	=VS_PROD_YR/12	=F38*VS_PCT_LIQUID

	A	B	H	I	J	K	L
23							
24		**MODEL:**					
25			**Emptying**	**VS Emptied**	**VS "Available"**	**VS "consumed"**	**CH4 Produced**
26		**Month**	***Y/N***	***kg***	***kg***	***kg***	***m3***
27		**1**	=VLOOKUP(B27,$C$9:$F$20,4,FALSE)	n/a	=G27	=J27*E27	=K27*B0
28		**2**	=VLOOKUP(B28,$C$9:$F$20,4,FALSE)	=IF(H28="N",0,(J27-K27)*EMPTY_EFFICIENCY)	=G28+J27-K27-I28	=J28*E28	=K28*B0
29		**3**	=VLOOKUP(B29,$C$9:$F$20,4,FALSE)	=IF(H29="N",0,(J28-K28)*EMPTY_EFFICIENCY)	=G29+J28-K28-I29	=J29*E29	=K29*B0
30		**4**	=VLOOKUP(B30,$C$9:$F$20,4,FALSE)	=IF(H30="N",0,(J29-K29)*EMPTY_EFFICIENCY)	=G30+J29-K29-I30	=J30*E30	=K30*B0
31		**5**	=VLOOKUP(B31,$C$9:$F$20,4,FALSE)	=IF(H31="N",0,(J30-K30)*EMPTY_EFFICIENCY)	=G31+J30-K30-I31	=J31*E31	=K31*B0
32		**6**	=VLOOKUP(B32,$C$9:$F$20,4,FALSE)	=IF(H32="N",0,(J31-K31)*EMPTY_EFFICIENCY)	=G32+J31-K31-I32	=J32*E32	=K32*B0
33		**7**	=VLOOKUP(B33,$C$9:$F$20,4,FALSE)	=IF(H33="N",0,(J32-K32)*EMPTY_EFFICIENCY)	=G33+J32-K32-I33	=J33*E33	=K33*B0
34		**8**	=VLOOKUP(B34,$C$9:$F$20,4,FALSE)	=IF(H34="N",0,(J33-K33)*EMPTY_EFFICIENCY)	=G34+J33-K33-I34	=J34*E34	=K34*B0
35		**9**	=VLOOKUP(B35,$C$9:$F$20,4,FALSE)	=IF(H35="N",0,(J34-K34)*EMPTY_EFFICIENCY)	=G35+J34-K34-I35	=J35*E35	=K35*B0
36		**10**	=VLOOKUP(B36,$C$9:$F$20,4,FALSE)	=IF(H36="N",0,(J35-K35)*EMPTY_EFFICIENCY)	=G36+J35-K35-I36	=J36*E36	=K36*B0
37		**11**	=VLOOKUP(B37,$C$9:$F$20,4,FALSE)	=IF(H37="N",0,(J36-K36)*EMPTY_EFFICIENCY)	=G37+J36-K36-I37	=J37*E37	=K37*B0
38		**12**	=VLOOKUP(B38,$C$9:$F$20,4,FALSE)	=IF(H38="N",0,(J37-K37)*EMPTY_EFFICIENCY)	=G38+J37-K37-I38	=J38*E38	=K38*B0

	A	B	F	G	I	J	K	L
23								
24		**MODEL:**						
25			*VS Excreted*	*VS Loaded*	**VS Emptied**	**VS "Available"**	**VS "consumed"**	**CH4 Produced**
26		**Month**	***kg/month***	***kg/month***	***kg***	***kg***	***kg***	***m3***
63								
64			=SUM(F27:F38)	=SUM(G27:G38)	=SUM(I27:I38)		=SUM(K27:K38)	=SUM(L27:L38)
65			=SUM(F39:F50)	=SUM(G39:G50)	=SUM(I39:I50)		=SUM(K39:K50)	=SUM(L39:L50)
66			=SUM(F51:F62)	=SUM(G51:G62)	=SUM(I51:I62)		=SUM(K51:K62)	=SUM(L51:L62)

Figure 5. Formulae used in the model. To conserve space, only 12 months are shown. Top panel: columns C:G. Middle panel: columns H:L. Bottom panel: sums in rows 64:66 for selected columns.



Figure 6. Monthly patterns in Year 3: manure temperature, VS available, VS emptied, and methane production.

MODEL RESULTS

The MCF is calculated in the Results section. This is done using the third year outputs. In this particular example, the input air temperature is from the Cool Temperate Moist region and the retention time is 6-months. The resulting MCF (21%) is identical with the guidance document (21%).



Figure 7. Summary of Year 3 VS and methane production, and calculation of MCF. Top panel shows results, bottom panel shows equations.

NOTE ABOUT TERMINOLOGY:

The terms “VS Available” and “VS Consumed” are used here to be consistent with IPCC 2006 and Mangino et al. 2001 approach. However, these terms require some clarification to avoid misinterpretation. **(1)** The term “VS Consumed” does not represent the reality of VS degradation. To put it simply, VS consumed is not equivalent to VS destroyed. Rather, the term VS Consumed can be thought of as the quantity of VS conceptually removed from the liquid/slurry storage and placed into a (conceptual) biomethane potential at 35°C (i.e. to produce the B_0). Therefore, just as B0 reports the quantity of CH_4 produced per kg of VS (i.e. all fractions, degradable and non-degradable), the concept of “VS Consumed” removes all fractions of VS from storage. This approach is convenient because it uses the B_0 as the integrator of all fractions of VS degradability, and reports the total methane produced from all fractions as if they were incubated for infinite time, while the *f* parameter introduces a temperature dependence. While this is convenient for modeling, and is consistent with the B_0, this is not really what is happening in a liquid/slurry storage. **(2)** Since “VS Consumed” does not equate with the amount of VS degraded in the storage, the “VS available” does equate with the amount of VS that would actually be measured in a storage. Therefore, researchers should not attempt to compare measured VS with “VS available”. **(3)** The strength of this approach is its simplicity and the fact that the maximum amount of methane that can be produced is equal to the total VS produced multiplied by the B_0. In other words, the model cannot produce more methane than the B_0. **(4)** The MCF is the ratio of predicted “VS Consumed” to the total VS that entered the storage over one year. The method does not address VS destruction. If the “VS Consumed” were multiplied by *B’* (m^3 CH4/kg VS destroyed), the result be would be erroneous because “VS Consumed” is not VS Destroyed. This is not to say that *B’* cannot be used to model methane production, but simply that it is not compatible with the “VS Consumed” concept. **(5)** Although B_0 does not need to enter the MCF calculation, the role of B_0 is to be multiplied by the MCF, as stated in equation 10.23 of IPCC 2006.